text: Title: Keyword timeout is not effective in teardown if keyword uses `Wait Until Keyword Succeeds`
  Body: When using `Wait Untless Keyword Succeeds` in test teardown, it is possible to have much longer execution time than allowed by the "user keyword timeout".
  
  In practice, in teardown, `Wait Untless Keyword Succeeds` does skip the keyword re-execution when timeout has been reached, but not the `retry_interval`.
  Out of teardown it works as expected, no re-execution occurs, therefore no `retry_interval` is included.
  
  Example code:
  ```robotframework
  *** Test Cases ***
  Test 1: Works as expected when used in test body
      [Documentation]    Does not last much longer than keyword timeout.
      My Keyword
  
  Test 2: Keyword timeout not respected in teardown
      [Documentation]    Lasts much longer than keyword timeout.
      No Operation
      [Teardown]    My Keyword
  
  
  *** Keywords ***
  My Keyword
      [Documentation]    This keyword should stop when reaching timeout
      [Timeout]    2s
      Wait Until Keyword Succeeds    4x    10s
      ...    Fail Slowly    60s
  
  Fail Slowly
      [Documentation]    Sleeps and fails.
      [Arguments]    ${sleep_duration}
      Sleep    ${sleep_duration}
      Fail    failed on purpose
  
  ```
  
  Example output:
  ![image](https://github.com/user-attachments/assets/de341070-a9e1-4088-a39c-6eeb92c6dfef)
  
  This feels like a bug when checked against documentation.
  - [Wait Until Keyword Succeeds states](https://robotframework.org/robotframework/latest/libraries/BuiltIn.html#Wait%20Until%20Keyword%20Succeeds):
  > All normal failures are caught by this keyword. Errors caused by invalid syntax, test or keyword timeouts, or fatal exceptions [...] are not caught.
  - User guide on [User keyword timeout](https://robotframework.org/robotframework/latest/RobotFrameworkUserGuide.html#user-keyword-timeout):
  > User keyword timeouts are applicable also during a test case teardown, whereas test timeouts are not.
  
  I believe the expected behavior would be that retry interval are also skipped when timeout has been reached.